

NAG Library Routine Document

S21CCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S21CCF returns the value of one of the Jacobian theta functions $\theta_0(x, q)$, $\theta_1(x, q)$, $\theta_2(x, q)$, $\theta_3(x, q)$ or $\theta_4(x, q)$ for a real argument x and non-negative $q < 1$, via the function name.

2 Specification

```
FUNCTION S21CCF (K, X, Q, IFAIL)
REAL (KIND=nag_wp) S21CCF
INTEGER           K, IFAIL
REAL (KIND=nag_wp) X, Q
```

3 Description

S21CCF evaluates an approximation to the Jacobian theta functions $\theta_0(x, q)$, $\theta_1(x, q)$, $\theta_2(x, q)$, $\theta_3(x, q)$ and $\theta_4(x, q)$ given by

$$\begin{aligned}\theta_0(x, q) &= 1 + 2 \sum_{n=1}^{\infty} (-1)^n q^{n^2} \cos(2n\pi x), \\ \theta_1(x, q) &= 2 \sum_{n=0}^{\infty} (-1)^n q^{\binom{n+1}{2}} \sin\{(2n+1)\pi x\}, \\ \theta_2(x, q) &= 2 \sum_{n=0}^{\infty} q^{\binom{n+1}{2}} \cos\{(2n+1)\pi x\}, \\ \theta_3(x, q) &= 1 + 2 \sum_{n=1}^{\infty} q^{n^2} \cos(2n\pi x), \\ \theta_4(x, q) &= \theta_0(x, q),\end{aligned}$$

where x and q (the *nome*) are real with $0 \leq q < 1$.

These functions are important in practice because every one of the Jacobian elliptic functions (see S21CBF) can be expressed as the ratio of two Jacobian theta functions (see Whittaker and Watson (1990)). There is also a bewildering variety of notations used in the literature to define them. Some authors (e.g., Section 16.27 of Abramowitz and Stegun (1972)) define the argument in the trigonometric terms to be x instead of πx . This can often lead to confusion, so great care must therefore be exercised when consulting the literature. Further details (including various relations and identities) can be found in the references.

S21CCF is based on a truncated series approach. If t differs from x or $-x$ by an integer when $0 \leq t \leq \frac{1}{2}$, it follows from the periodicity and symmetry properties of the functions that $\theta_1(x, q) = \pm\theta_1(t, q)$ and $\theta_3(x, q) = \pm\theta_3(t, q)$. In a region for which the approximation is sufficiently accurate, θ_1 is set equal to the first term ($n = 0$) of the transformed series

$$\theta_1(t, q) = 2\sqrt{\frac{\lambda}{\pi}} e^{-\lambda t^2} \sum_{n=0}^{\infty} (-1)^n e^{-\lambda \left(n + \frac{1}{2}\right)^2} \sinh\{(2n + 1)\lambda t\}$$

and θ_3 is set equal to the first two terms (i.e., $n \leq 1$) of

$$\theta_3(t, q) = \sqrt{\frac{\lambda}{\pi}} e^{-\lambda t^2} \left\{ 1 + 2 \sum_{n=1}^{\infty} e^{-\lambda n^2} \cosh(2n\lambda t) \right\},$$

where $\lambda = \pi^2 / |\log_e q|$. Otherwise, the trigonometric series for $\theta_1(t, q)$ and $\theta_3(t, q)$ are used. For all values of x , θ_0 and θ_2 are computed from the relations $\theta_0(x, q) = \theta_3\left(\frac{1}{2} - |x|, q\right)$ and $\theta_2(x, q) = \theta_1\left(\frac{1}{2} - |x|, q\right)$.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Byrd P F and Friedman M D (1971) *Handbook of Elliptic Integrals for Engineers and Scientists* pp. 315–320 (2nd Edition) Springer–Verlag

Magnus W, Oberhettinger F and Soni R P (1966) *Formulas and Theorems for the Special Functions of Mathematical Physics* 371–377 Springer–Verlag

Tölke F (1966) *Praktische Funktionenlehre (Bd. II)* 1–38 Springer–Verlag

Whittaker E T and Watson G N (1990) *A Course in Modern Analysis* (4th Edition) Cambridge University Press

5 Parameters

- 1: K – INTEGER *Input*
On entry: denotes the function $\theta_k(x, q)$ to be evaluated. Note that $K = 4$ is equivalent to $K = 0$.
Constraint: $0 \leq K \leq 4$.
- 2: X – REAL (KIND=nag_wp) *Input*
On entry: the argument x of the function.
- 3: Q – REAL (KIND=nag_wp) *Input*
On entry: the argument q of the function.
Constraint: $0.0 \leq Q < 1.0$.
- 4: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $K < 0$,
or $K > 4$,
or $Q < 0.0$,
or $Q \geq 1.0$,

$IFAIL = 2$

The evaluation has been abandoned because the function value is infinite. The result is returned as the largest machine representable number (see $X02ALF$).

7 Accuracy

In principle the routine is capable of achieving full relative precision in the computed values. However, the accuracy obtainable in practice depends on the accuracy of the standard elementary functions such as \sin and \cos .

8 Further Comments

None.

9 Example

This example evaluates $\theta_2(x, q)$ at $x = 0.7$ when $q = 0.4$, and prints the results.

9.1 Program Text

```

Program s21ccfe

!      S21CCF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, s21ccf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: q, x, y
      Integer                      :: ifail, k
!      .. Executable Statements ..
      Write (nout,*) 'S21CCF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Write (nout,*)
      Write (nout,*) ' K      X      Q      Y'
      Write (nout,*)

      Read (nin,*) k, x, q

      ifail = -1
      y = s21ccf(k,x,q,ifail)

```

```
      If (ifail>=0) Then
        Write (nout,99999) k, x, q, y
      End If

99999 Format (1X,I2,2X,F4.1,2X,F4.1,2X,1P,E12.4)
      End Program s21ccfe
```

9.2 Program Data

S21CCF Example Program Data
2 0.7 0.4 : Values of K, X and Q

9.3 Program Results

S21CCF Example Program Results

K	X	Q	Y
2	0.7	0.4	-6.9289E-01
